


ODS Graphics for the Non-Statistician

By Mike Kalt and Cynthia Zender*

*Updated June 2022

Copyright © SAS Institute Inc. All rights reserved.



1

Agenda

- What is ODS Graphics?
- Using Base SAS procedures to produce graphics
- Overview of SAS/GRAPH "SG" procedures
- Using PROC SGPLOT *
- Controlling graph appearance *
- Annotation *
- Using SGPIE *
- Using SGMAP *
- Using SGPANEL

Refer to original paper for these topics: Using SGSCATTER, Using the ODS Graphics Designer, and Editing graphs produced by SG procedures

<https://support.sas.com/resources/papers/proceedings11/294-2011.pdf>

2

Copyright © SAS Institute Inc. All rights reserved.

2

What Is ODS GRAPHICS?



3


Copyright © SAS Institute Inc. All rights reserved.



3

In the Beginning. . .

All SAS graphics were done with "classic" SAS/GRAPH procedures such as PROC GCHART and PROC GPLOT



4


Copyright © SAS Institute Inc. All rights reserved.



4

ODS Graphics: Why?

Designed to make it easier for statistical users to develop commonly-used statistical graphics



5

Copyright © SAS Institute Inc. All rights reserved.



5

ODS Graphics: What is It?

- Graphics capabilities added to statistical procedures
- Graphics capabilities added to some Base SAS procedures
- Part of Base SAS license:
 - "SG" procedures
 - Graphics Template Language
 - ODS Graphics Designer



This collection of features is **ODS Graphics**.

6


Copyright © SAS Institute Inc. All rights reserved.



6

Not Just for Statisticians!

Although the original purpose of ODS Graphics was to make producing statistical graphics easier, it's also a convenient way for producing general purpose plots and charts.

$$S = \left(\frac{1}{N-1} \sum_{i=1}^N (y_i - \bar{y})^2 \right)^{\frac{1}{2}}$$


7

Copyright © SAS Institute Inc. All rights reserved.



7

ODS Graphics: How to Use It

For statistical and Base SAS procedures

- Use ODS GRAPHICS statement to activate graphics
- Add options to procedure code to generate graphs
- Use ODS GRAPHICS OFF statement to deactivate graphics

For SG Procedures

- Use SG procedure statements
- No ODS GRAPHICS statement required but may be needed to change defaults

8

Copyright © SAS Institute Inc. All rights reserved.



8

What's New


To see what was new in each SAS 9.4 release, search for this string:



site: support.sas.com "What's New in SAS ODS Graphics Procedures 9.4"

site: support.sas.com "What's New in SAS ODS Graphics Procedures 9.4"

On Google the top hit is:



9




Copyright © SAS Institute Inc. All rights reserved.


9

ODS Graphics Example: Base/Stat Procedures

```
ods graphics on;
proc freq data=orion.back_orders;
  tables region/plots=freqplot;
run;
ods graphics off;
```



s01_demo3_freq-example.sas



10




Copyright © SAS Institute Inc. All rights reserved.

10

ODS Graphics Example: "SG" Procedure

```
proc sgplot data=orion.back_orders;
  vbar region;
  title 'Back Orders by Region';
run;
```



s01_demo4_sgplot_vbar.sas

11

Copyright © SAS Institute Inc. All rights reserved.



11

ODS Graphics: Output

Output to image file**

- ODS LISTING destination used
- Image file accessed from Results window
- Output displayed in host image viewer

Output to an ODS destination**

- Specify destination on ODS statement
- Graph is output as an image in the ODS document
- Output document can be viewed in Results Viewer or in rendering application (such as Word, Acrobat Reader)

** Depends on the interface used, for SAS Studio and Enterprise Guide, default output is to HTML or HTML5.

12

Copyright © SAS Institute Inc. All rights reserved.



12

SAS/GRAPH "Classic" vs. ODS Graphics

SAS/GRAPH Classic Procedures	ODS Graphics
Output goes to GRAPH1 window or other destinations (including ODS)	Output to image file or ODS document only
Graph created as a GRSEG entry in a SAS catalog	No catalog entries created
GREPLAY procedure replays graphs stored in catalogs	No GREPLAY procedure
Annotate facility available to add elements to existing graphs	Annotate facility available to add elements to existing graphs
GOPTIONS statement sets general graphics options	ODS GRAPHICS statement sets general graphics options (GOPTIONS statement not used)

Output from "Classic" SAS/GRAPH procedures cannot be combined in SAS with output from ODS Graphics.

13

Copyright © SAS Institute Inc. All rights reserved.




13

Base SAS Procedures that Produce Graphics

ODS Graphics can be used with the following Base SAS procedures:

- CORR
- UNIVARIATE
- FREQ

```
ods graphics on;
proc freq data=orion.employee_payroll;
  tables marital_status
    /plots=freqplot
      (orient=horizontal
       scale=percent);
run;
ods graphics off;
```



14

s02_demo4_freq-plot-options-2.sas



14

ODS GRAPHICS Procedures and Features

The ODS GRAPHICS procedures and features are:

- PROC SGPLOT
- PROC SGPANEL
- PROC SGSCATTER
- PROC SGMAP (SAS 9.4M6) (along with SAS/GRAPH procedures needed with maps such as GPROJECT, GREduce, etc)
- PROC SGPIE (pre-production in SAS 9.4M7)
- SG Annotate feature (SAS 9.3)
- SG Attribute Maps (SAS 9.4M3)

15

Copyright © SAS Institute Inc. All rights reserved.



15

But WAIT...There's More

Other procedures and template language are also part of ODS GRAPHICS. They are:

- ODS GRAPHICS DESIGNER
- PROC SGDESIGN
- PROC SGRENDER
- Graphics Template Language (GTL)

16

Copyright © SAS Institute Inc. All rights reserved.



16

The "SG" Procedures

SGPLOT, SGPanel, SGSCATTER, SGMAP, SGPIE Procedures

- designed to create commonly used graphs quickly.
- produce many of the same types of graphs as the original SAS/GRAPH procedures such as GPLOT and GCHART, but use the ODS architecture

SGRENDER, SGDESIGN, ODS GRAPHICS Designer and the Graphics Template Language

- used to produce more complex, customized graphs
- steeper learning curve
- not covered in this presentation

17

Copyright © SAS Institute Inc. All rights reserved.



17

The SGPLOT Procedure

PROC SGPLOT can be used to produce plots and charts, such as the following:

- Simple bar charts
- Stacked bar charts
- Histograms
- Scatter plots
- Series plots
- Overlaid graphs including bar/line charts
-and many more

18


Copyright © SAS Institute Inc. All rights reserved.



18

Terminology for ODS Graphics

Term	Meaning
plot	any type of plot or chart, such as a scatter plot, bar chart, etc.
cell	area containing one plot, or multiple overlaid plots
graph	a collection of one or more cells




SAS

19

Vertical Bar Chart

```
proc sgplot data=orion.back_orders;
  vbar region;
  title "Frequency Count by Region";
run;
```




s04_demo1_sgplot-vbar.sas

SAS

20

Horizontal Bar Charts

```
proc sgplot data=orion.back_orders;
  hbar region / response=number_of_orders stat=sum;
  title "Backorders by Region";
run;
```



s05_demo10_ods-graphics-statement.sas

21


Copyright © SAS Institute Inc. All rights reserved.



21

Subgrouped Bar Chart

```
proc sgplot data=orion.employees;
  where company=: 'Orion';
  vbar company / group=department;
  title 'Employees By Country/Department';
run;
```




s04_demo3_sgplot-vbar-subgroup.sas



22

Histogram

```
proc sgplot data=orion.employees;
  where department='Sales';
  histogram salary;
  title 'Distribution of Employee Salaries';
run;
```




s04_demo4_sgplot-histogram.sas



23

Overlaying Graphs

```
proc sgplot data=orion.employees;
  histogram salary;
  density salary / type=normal;
  where department='Sales';
  title 'Distribution of Employee Salaries';
run;
```




s04_demo5_sgplot-histogram-density.sas



24

Watch Statement Order

```
proc sgplot data=orion.employees;
title 'Distribution of Employee Salaries';
  where department='Sales';
  density salary / type=normal;
  histogram salary;
run;
```




s04_demo5_sgplot-histogram-density.sas



25

Scatter Plot

```
proc sgplot data=orion.employees;
where company='Orion Italy' and department='Sales';
  scatter y=salary x=employee_birthdate;
  title 'Salary by Birth Date';
run;
```




s04_demo6_sgplot-scatter.sas



26

Scatter Plot with Band

```
proc sgplot data=orion.employees;
where company='Orion Italy' and department='Sales';
band x=employee_birthdate lower=26000 upper=30500/legendlabel='In Range';
scatter y=salary x=employee_birthdate;
title 'Salary by Birth Date';
run;
```



s04_demo6_sgplot-scatter.sas


sas

Copyright © SAS Institute Inc. All rights reserved.

27

Grouped Series Plot

```
proc sgplot data=orion.profit;
where company in ('Orion France', 'Orion Italy');
series y=sales x=yymm / group=company;
title 'Sales by Country';
run;
```



s04_demo7_sgplot-series-grouped.sas

sas

Copyright © SAS Institute Inc. All rights reserved.


28

Overlay Series Plot with Multiple Y Variables

```
proc sgplot data=orion.budget;
  series y=yr2003 x=month;
  series y=yr2004 x=month;
  title 'Budget by Month';
run;
```

NOTE: Month is a numeric value from 1-12. It is treated as a continuous value by default.

s04_demo8_sgplot-series-overlay.sas




29

Overlay Series Plot with Multiple Y Variables

```
proc sgplot data=orion.budget;
  series y=yr2003 x=month / lineattrs=(color=green pattern=1);
  series y=yr2004 x=month / lineattrs=(color=brown pattern=4);
  xaxis type=discrete;
  title 'Budget by Month';
run;
```

NOTE: Use TYPE=DISCRETE to change the tick values on the X axis.


s04_demo8_sgplot-series-overlay.sas



30

Overlaid Bar Chart and Line Plot

```
proc sgplot data=orion.profit;
  where company='Orion France' and year(yymm)=2004;
  vbar yymm / response=sales;
  vline yymm / response=cost;
  title 'Sales vs. Cost';
run;
```



s04_demo9_sgplot-bar-line.sas

sas

31

Using ODS Styles

```
ods html style=banker;
proc sgplot data=orion.profit;
  where company='Orion France' and year(yymm)=2004;
  vbar yymm / response=sales;
  vline yymm / response=cost;
  title 'Sales vs. Cost';
run;
```




s05_demo01_sgplot-style.sas

sas

32

Controlling Fills and Labeling Bars

```
proc sgplot data=orion.back_orders;
  vbar region/response=number_of_orders
    fillattrs=(color="verylightred")
    datalabel;
  title "Back Orders by Region";
run;
```




s05_demo02_sgplot-vbar-fill-label.sas



33

Overlaid Bars with Transparency

```
proc sgplot data=orion.budget;
  title 'Budget by Month';
  vbar month/response=yr2003 fillattrs=(color=red) transparency=.7;
  vbar month/response=yr2004 fillattrs=(color=blue) transparency=.7
    barwidth=.5;
run;
```



s05_demo03_sgplot_vbar_transparency.sas




Copyright © SAS Institute Inc. All rights reserved.

34

Controlling Lines and Markers

```
proc sgplot data=orion.budget;
  series y=yr2003 x=month/ markers
    lineattrs=(color=red thickness=3)
    markerattrs=(color=black symbol=circlefilled size=8);
  title 'Budget by Month';
run;
```




s05_demo04_sgplot-series-attributes.sas



35

Controlling Axis Attributes

```
proc sgplot data=orion.profit;
  where (company='Orion Australia' or
    company='Orion Italy') and year(yymm)=2003;
  series y=sales x=yymm/ markers group=company;
  xaxis fitpolicy=rotate tickvalueformat=monname. grid label='Month';
  yaxis label='Total Sales' grid;
  title '2003 Sales';
run;
```



s05_demo05_sgplot-series-axis.sas




Copyright © SAS Institute Inc. All rights reserved.

36

Adding Top and Right Axes

```
proc sgplot data=orion.budget;
  series y=yr2003 x=month;
  series y=yr2004 x=month/x2axis y2axis;
  xaxis type=discrete;
  yaxis display=(nolabel) grid offsetmax=.1;
  x2axis type=discrete display=(nolabel);
  y2axis display=(nolabel) values=(1e6 to 3e6 by 500000)
    offsetmax=.1;
  title 'Budget by Month';
run;
```

s05_demo07_sgplot-series-axis2.sas




sas

37

Controlling the Legend

```
proc sgplot data=orion.budget;
  series y=yr2003 x=month;
  series y=yr2004 x=month;
  xaxis type=discrete;
  keylegend / location=inside down=2 position=topleft;
  title 'Budget by Month';
run;
```

s05_demo08_sgplot-series-keylegend.sas



sas

38

Using the INSET Statement

```
proc sgplot data=orion.clinic_stats;
  title 'Vbar and Vline with Inset';
  vbar monyy / response=tot_pts group=clinic fillattrs=(transparency=0.25);
  vline monyy / response=num_surg y2axis group=clinic lineattrs=(thickness=3px);
  inset (" Minimum" = "&minsurg"
        " Mean" = "&avgsurg"
        " Max" = "&maxsurg") /
        title='Clinic Surgeries'
        border position=top;
  xaxis label = 'Month' interval=month fitpolicy=stagger;
  yaxis label='Total Patients' values=(0 to 2500 by 500);
  y2axis label = 'Number Surgeries' values=(0 to 100 by 10);
run;
```

s05_demo09_sgplot-all-inset.sas

39


Macro variables **&minsurg**, **&avgsurg** and **&maxsurg** were created using PROC MEANS and DATA_NULL_steps.

39

Using the INSET Statement

```
proc sgplot data=orion.clinic_stats;
  title 'Vbar and Vline with Inset';
  vbar monyy / response=tot_pts group=clinic fillattrs=(transparenc
  vline monyy / response=num_surg y2axis group=clinic lineattrs=(thicknes
  inset (" Minimum" = "&minsurg"
        " Mean" = "&avgsurg"
        " Max" = "&maxsurg") /
        title='Clinic Surgeries'
        border position=top;
  xaxis label = 'Month' interval=month fitpolicy=stagger;
  yaxis label='Total Patients' values=(0 to 2500 by 500);
  y2axis label = 'Number Surgeries' values=(0 to 100 by 10);
run;
```

Macro variables **&minsurg**, **&avgsurg** and **&maxsurg** were created using PROC MEANS and DATA_NULL_steps.



40

The ODS GRAPHICS Statement

You can specify options on the ODS GRAPHICS statement to control some aspects of your graphics output. Some of the options are:

- Height and width of the graph
- Format (file type) and name of the graphics file
- Whether a border is drawn around the graph

Options specified on the ODS GRAPHICS statement remain in effect until they are reset or the SAS session ends. To reset the options, submit the following statement:

```
ods graphics/reset;
```

41




Copyright © SAS Institute Inc. All rights reserved.

41

Using the ODS GRAPHICS Statement

```
ods graphics / height=200px width=500px imagefmt=jpeg;
proc sgplot data=orion.back_orders;
  vbar region/response=number_of_orders
    fillattrs=(color="verylightred")
    datalabel;
title "Back Orders by Region";
run;
```



s05_demo10_ods-graphics-statement.sas

42



Copyright © SAS Institute Inc. All rights reserved.

42


Adding a Table to a Graph with XAXISTABLE

```
%let region = United States;
title "1) Total Sales for &region by Subsidiary";
proc sgplot data=shoes;
  vbarbasic subsidiary /response=sales group=subsidiary;
  xaxistable sales / x=subsidiary class=product;
  axis display=(nolabel noline);
  yaxis display=(noline) grid;
run;
```

Note:

VBARBASIC statement added 9.4M3
XAXISTABLE, YAXISTABLE added 9.4M1

s11_d1_vbarbasic_xaxistable.sas



Copyright © SAS Institute Inc. All rights reserved.

as

43

VBAR, VBARPARAM and VBARBASIC

The SGPLOT statements VBAR, VBARPARAM and VBARBASIC all produce vertical bar charts.

- VBAR produces a basic vertical bar chart on non-summarized data
- VBARPARAM produces a basic vertical bar chart on pre-summarized data
- VBARBASIC is a categorical type of plot statement that allows you to combine vertical bar charts using non-summarized data with other basic plot types. (starting in 9.4M3)

Other SGPLOT statements have similar plot types, such as HBAR, HBARPARAM and HBARBASIC. Refer to the SGPLOT documentation or the Graphically Speaking blog for more usage examples.

44

Copyright © SAS Institute Inc. All rights reserved.

sas

44

Using YAXISTABLE


```
proc sgplot data=sashelp.class (where=(sex='F')) noautolegend;
  bubble x=height y=name size=weight / group=name datalabel=name
  transparency=0.4 datalabelattrs=(size=4pt weight=bold);
  inset "Bubble size represents weight"
  position=bottomright textsize=8;
  yaxis grid;
  yaxistable age height weight;
run;
```

Note:

BUBBLE statement added 9.3

XAXISTABLE, YAXISTABLE added 9.4M1

s11_d2_bubble_yaxistable.sas




45

Pie for Dessert!

```
proc format;
  value $type
    "Hybrid", "Sedan" = "Passenger Cars",
    "SUV", "Wagon" = "Larger Cars",
    "Truck", "Sports" = "Sports/Trucks";
run;
```


```
/*--PIE-- SAS 9.4M6 (preproduction)
ods graphics / reset width=6in;
title '1) Vehicle Distribution';
proc sgpie data=sashelp.cars;
  format type $type.;
  styleattrs datacolors=(green blue purple);
  pie type / startangle=270
  dataskin=gloss;
run;
```

s11_d3_sgpie_donut.sas



46

How To Ensure Consistent Colors



47

Using Attribute Maps

```
data attrmap;
  length value fillcolor linecolor $10;
  infile datalines dlm=' ';
  input value $ fillcolor $;
  id='barcolors';
  linecolor=fillcolor;
datalines;
Strawberry,pink
Chocolate,CX7B3F00
Vanilla,beige
;
run;
```

AttrMap Table				
Obs	value	fillcolor	linecolor	id
1	Strawberry	pink	pink	barcolors
2	Chocolate	CX7B3F00	CX7B3F00	barcolors
3	Vanilla	beige	beige	barcolors

s11_d4_icecream_attrmap.sas

48

Copyright © SAS Institute Inc. All rights reserved.

48

Link Attribute Map to SGPLOT

```

title "2) Ice Cream Survey for Group #byval(grp)";
proc sgplot data=icecream dattrmap=attrmap
            noautolegend;
  by grp;
  vbar flavor / response=rank group=flavor
              dataskin=pressed attrid=barcolors;
run;

```

AttrMap Table



Obs	value	fillcolor	linecolor	id
1	Strawberry	pink	pink	barcolors
2	Chocolate	CX7B3F00	CX7B3F00	barcolors
3	Vanilla	beige	beige	barcolors

s11_d4_icecream_attrmap.sas

DATTRMAP option specifies attribute map dataset.
ATTRID option specifies ID value for this variable (Flavor).

49

Copyright © SAS Institute Inc. All rights reserved.



49

Using Hex Symbols with SYMBOLCHAR

```


proc sgplot data=temperatures dattrmap=myattrmap;
  highlow x=day low=low high=high;
  scatter x=day y=high / marker=highlow;
  symbolchar name=Cloud char='☁';
  symbolchar name=Sun char='☀';
  symbolchar name=Umbrella char='☔';
  xaxis label='Day of the Week' / valueattrs=(font=helvetica);
  xaxis table high / valueattrs=(font=helvetica);
  xaxis table low / valueattrs=(font=helvetica);
run;

```

Attribute Map for Marker Symbol

Obs	id	value	markercolor	markersymbol
1	myid	Cloudy	grey	Cloud
2	myid	Sunny	orange	Sun
3	myid	Showers	blue	Umbrella

s11_d5_symbolchar.sas



50

Create Annotation Dataset

Annotate Dataset: work.annotable

Obs	x1space	y1space	anchor	year	chicken	beef	pork	width	widthunit	function	x1	y1	image	textsize	label
1	datapercnt	datavalue	left	2000	77	67	51	40	pixel	image	102	77	C:\ods_graf_non_stat\odsgns\output\chicken.jpg	.	
2	datapercnt	datavalue	left	2000	77	67	51	40	pixel	image	102	67	C:\ods_graf_non_stat\odsgns\output\cow.jpg	.	
3	datapercnt	datavalue	left	2000	77	67	51	40	pixel	image	102	51	C:\ods_graf_non_stat\odsgns\output\pig.jpg	.	
4	graphpercent	graphpercent	bottomright	2000	.	.	.	90	pixel	image	99	1	C:\ods_graf_non_stat\odsgns\output\Logo.png	.	
5	graphpercent	graphpercent	bottomleft	2000	.	.	.	150	pixel	text	1	1		6	Source: USDA

Position on Y axis

51

Copyright © SAS Institute Inc. All rights reserved.




51

With Annotation

```
proc sgplot data=meat_consumption sganno=work.annotable
    noautolegend pad=(bottom=8%);
    xaxis display=(nolabel) offsetmax=0.1;
    yaxis label="Retail Cut Equivalent / Lb. per Person";
    series x=year y=chicken / marker=triangle;
    scatter x=year y=chicken / marker=triangle;
    series x=year y=pork / marker=triangle;
    scatter x=year y=pork / marker=triangle;
    series x=year y=beef / ..;
    scatter x=year y=beef / marker=triangle;
run;
```

s10_demo5_annotate_image.sas



52

SGMAP Procedure

Beginning in SAS version 9.4 M5, the SGMAP procedure allows you to create maps and then overlay plots such as text, scatter, or bubble plots. There are 3 different types of mapping statements used with the SGMAP procedure.

- 1) **Data:** PROC SGMAP has 3 data related options PLOTDATA=, MAPDATA=, and MAPRESPDATA=.
- 2) **Map background:** PROC SGMAP has 3 possible map background statements. Use one of these statements to define map background OPENSTREETMAP, ESRIMAP, or CHOROMAP.
- 3) **Overlay:** Use BUBBLE, SCATTER, TEXT, and SERIES statements to specify the plots you want to overlay on the map background.

The map appearance can be customized by changing colors, fonts, size and position of many graph elements.

53

Copyright © SAS Institute Inc. All rights reserved.



53

Using Other Procedures to Prepare Your Data

Before you create your map, you may have to use other mapping related procedures to help you prepare your data.

- GEOCODE: Codes geographic coordinates.
- GINSIDE: Limits data points to the ones that fall inside the map area of interest.
- GPROJECT: Converts spherical coordinates (LAT, LONG) onto flat plane for mapping (Cartesian coordinates).
- GREduce: Simplifies map boundaries and creates a map dataset with a DENSITY variable.
- GREMOVE: Removes internal boundaries between map areas.
- MAPIMPORT: Imports shapefiles to create an output dataset for SGMAP or GMAP.

54

Copyright © SAS Institute Inc. All rights reserved.



54

Simple Example SGMAP: CHOROMAP

```
title 'CHOROMAP with Divisions color coded';
proc sgmap mapdata=mapsgfk.us
    maprespdata=sashelp.us_data;
    choromap Division / id=statecode mapid=statecode;
run;
```

- The **MAPSGFK** library contains new map data sets that are licensed through GfK Geomarketing. These maps are updated with the most current map data.
- The **MAPSGFK** data has better mapping variables. The **X** and **Y** variables contain projected values. The **LONG** and **LAT** variables contain the unprojected values.
- The **MAPS** and **MAPSSAS** libraries contain the older SAS/GRAPH map data sets. This library is NOT updated. You might want to consider moving legacy code to use the MAPSGFK data if using current map data is necessary.

Find ODS Tip sheet here:

https://support.sas.com/rnd/datavisualization/papers/tipsheets/SGF2019_GfKMapData_TipSheet.pdf

s11_d6_sgmap_examples.sas

55

Copyright © SAS Institute Inc. All rights reserved.




55

Simple Example SGMAP: CHOROMAP

```
title 'CHOROMAP with
proc sgmap mapdata=ma
    maprespdat
    choromap Divis
run;
```

s11_d6_sgmap_examples.sas



56

OPENSTREETMAP and SCATTER Example

```
PROC SGMAP PLOTDATA=eastUSsprings;
  TITLE H=2 'Selected Springs East of the Mississippi';
  OPENSTREETMAP;
  SCATTER X=longitude Y=latitude / group=Type datalabel=dlabel
          DATALABELATTRS=(color=black weight=bold size=8pt)
          datalabelpos=top
          MARKERATTRS=(SIZE=5 SYMBOL=circlefilled);
RUN;
```

s11_d6_sgmap_examples.sas

57

Copyright © SAS Institute Inc. All rights reserved.



57

OPENSTREETMAP and SCATTER Example

```
PROC SGMAP PLOTDATA=eastUSsprings;
  TITLE H=2 'Selected Springs East of the Mississippi';
  OPENSTREETMAP;
  SCATTER X=longitude Y=latitude / group=Type datalabel=dlabel
          DATALABELATTRS=(color=black weight=bold size=8pt)
          datalabelpos=top
          MARKERATTRS=(SIZE=5 SYMBOL=circlefilled);
RUN;
```

s11_d6_sgmap_examples.sas

Selected Springs East of the Mississippi



58

OPENSTREETMAP and BUBBLE Example

```
PROC SGMAP PLOTDATA=eastUSsprings;
  TITLE H=2 'Selected Springs East of the Mississippi';
  OPENSTREETMAP;
  bubble X=longitude Y=latitude size=Fahrenheit /
    bradiusmin=5px bradiusmax=15px
    group=Type datalabel=dlabel datalabelpos=top
    datalabelattrs=(color=black weight=bold size=8pt);
RUN;
```

s11_d6_sgmap_examples.sas

59

sas

Copyright © SAS Institute Inc. All rights reserved.

59

OPENSTREETMAP and BUBBLE Example

```
PROC SGMAP PLOTDATA=eastUSsprings;
  TITLE H=2 'Selected Springs East of the Mississippi';
  OPENSTREETMAP;
  bubble X=longitude Y=latitude size=Fahrenheit /
    bradiusmin=5px bradiusmax=15px
    group=Type datalabel=dlabel datalabelpos=top
    datalabelattrs=(color=black weight=bold size=8pt);
RUN;
```

s11_d6_sgmap_examples.sas



60

Complex ESRIMAP Example

```
proc mapimport
  datafile='dorian_shapefile/al052019-040_5day_lin.shp'
  out=line;
  id stormname;
run;
data line; set line (rename=(x=long y=lat));
run;

proc sgmap mapdata=cone maprespdata=cone_attr plotdata=line noautolegend;
  esrimap url="http://services.arcgisonline.com/arcgis/rest/services/NatGeo_World_Map";
  choromap colorvar / mapid=stormname lineattrs=(color=orange thickness=4px) transparency=.7;
  series x=long y=lat / lineattrs=(color=firebrick thickness=1px) smoothconnect;
  scatter x=long y=lat / markerattrs=(color=firebrick size=20px symbol=circlefilled) transparency=.5;
run;
```

Search on the Graphically Speaking Blog for Robert Allison's Hurricane Dorian example on how to plot hurricane data on a map.

61

sas

Copyright © SAS Institute Inc. All rights reserved.

61

ESRIMAP Map Background




sas

62

PROC SGPANEL

PROC SGPANEL produces graphs similar to PROC SGPLOT. The SGPANEL output contains separate cells for each value of a categorical variable or crossing of multiple categorical variables. You can control the placement of the cells on a graph.



SAS

63

General SGPANEL Syntax

- PROC SGPANEL statement—invokes the procedure and specifies input data set
- Plot statement—specifies type of graph, variables, and options. The plot statements supported by PROC SGPANEL are the same as those supported by PROC SGPLOT. As with PROC SGPLOT, multiple plot statements produce overlay graphs.
- PANELBY statement—specifies the classification variables and the arrangement of cells on the page.
- Axis statements—control appearance of axes (optional)
- KEYLEGEND statement—controls appearance of legend (optional)
- Appearance options — The same options used to control plot appearance with PROC SGPLOT (such as LINEATTRS, FILLATTRS, etc.) can be used with PROC SGPANEL.

64

Copyright © SAS Institute Inc. All rights reserved.


SAS

64

The PANELBY Statement

The PANELBY statement lists one or more categorical variables.

```
proc sgpanel data=orion.profit;
where company in ('Orion Belgium' 'Orion Germany'
                 'Orion France' 'Orion Italy');
panelby company;
series y=sales x=yymm;
title 'Monthly Sales';
run;
```



s06_demo1_sgpanel-default.sas

SAS


Copyright © SAS Institute Inc. All rights reserved.

65

PANELBY Statement Options

Options on the PANELBY statement specify the arrangement of cells

```
proc sgpanel data=orion.profit;
where company in ('Orion Belgium' 'Orion Germany'
                 'Orion France' 'Orion Italy');
panelby company/layout=columnlattice onepanel;
series y=sales x=yymm;
title 'Monthly Sales';
run;
```



s06_demo2_sgpanel-columnlattice.sas


SAS

Copyright © SAS Institute Inc. All rights reserved.

66

Multiple PANELBY Variables

```
proc sgpanel data=orion.customer_orders;
title 'Orders by Country, Order Type, Gender';
panelby order_type customer_gender;
vbar customer_country;
run;
```




s06_demo4_sgpanel_multiple_vars.sas

sas

67

Removing Borders Between Cells

```
proc sgpanel data=orion.customer_orders;
title 'Orders by Country, Gender';
panelby customer_gender / noborder spacing=30 novarname;
vbar customer_country;
run;
```



s06_demo5_sgpanel_noborder.sas

sas

68

SGPANEL and ROWAXISTABLE

```

title 'SGPANEL and ROWAXISTABLE';
proc sgpanel data=sashelp.class (where=(age > 13));
  panelby sex / layout=rowlattice uniscale=column;
  hbar name / response=height stat=mean nostatlabel;
  rowaxis label='Height';
  rowaxistable age / stat=mean label='Age'
    valueattrs=(color=navy weight=bold)
    labelattrs=(color=navy weight=bold);
  rowaxistable weight / stat=mean label='Weight'
    valueattrs=(color=navy weight=bold)
    labelattrs=(color=navy weight=bold);
run;

```

ROWAXISTABLE, COLAXISTABLE added 9.4M3;

s06_demo7_sgpanel_rowaxistable.sas

69

SAS

Copyright © SAS Institute Inc. All rights reserved.

69

SGPANEL and ROWAXISTABLE

```

title 'SGPANEL and ROWAXISTABLE';
proc sgpanel data=sashelp.class (where=(age > 13));
  panelby sex / layout=rowlattice uniscale=column;
  hbar name / response=height stat=mean nostatlabel;
  rowaxis label='Height';
  rowaxistable age / stat=mean label='Age'
    valueattrs=(color=navy weight=bold)
    labelattrs=(color=navy weight=bold);
  rowaxistable weight / stat=mean label='Weight'
    valueattrs=(color=navy weight=bold)
    labelattrs=(color=navy weight=bold);
run;


```

ROWAXISTABLE, COLAXISTABLE statements and LABEL= option added 9.4M3;

s06_demo7_sgpanel_rowaxistable.sas

70

Copyright © SAS Institute Inc. All rights reserved.



70

More Resources

Graphically Speaking blog – Visual Index

<https://support.sas.com/rnd/datavisualization/graphicallyspeakingindex/>

Robert Allison web site:

<https://www.robslink.com/SAS/Home.htm>

Lex Jansen web site:

<https://www.lexjansen.com>

SAS Community Forum for Graphics Programming:

https://communities.sas.com/t5/Graphics-Programming/bd-p/sas_graph

71

Copyright © SAS Institute Inc. All rights reserved.



71

Questions?

Copyright © SAS Institute Inc. All rights reserved.



72